Создал некоторую Таблицу значений, которую нужно перевести в ДеревоЗначений и определить уровни. Не доконца понимаю какие методы использовать для заполнения, пытался циклом обойти Таблицу и заполнить Дерево, но отладкой прошелся не добавляет.
---
Я решил задачу, вопрос было только понять синтаксис уровней как задается, почему то некоторые люди неадекватно начинают отвечать имея опыт за плечами.
---
какие методы использовать для заполнения
Методы Добавить(), Найти().
---
```1c
    ДеревоЗначений = ТаблицаЗначенийВДеревоЗначений(КаталогТоваров, "Категория")
```

```1c
    Функция ТаблицаЗначенийВДеревоЗначений(ТЗ, ПоляГруппировок)
    Группировки = СтрРазделить(ПоляГруппировок, ",", Ложь);
    ДеревоЗначений = Новый ДеревоЗначений;

    Для Каждого Колонка Из ТЗ.Колонки Цикл
    ДеревоЗначений.Колонки.Добавить(Колонка.Имя, Колонка.ТипЗначения);
    КонецЦикла;

    СоздатьСтрокиДереваРекурсивно(ТЗ, ДеревоЗначений.Строки, Группировки);
    Возврат ДеревоЗначений;
    КонецФункции

    Процедура СоздатьСтрокиДереваРекурсивно(ТаблицаЗначений, СтрокиДерева, Группировки, ИндексГруппировки = 0)

    Если ИндексГруппировки = Группировки.Количество() Тогда

    Для Каждого Стр Из ТаблицаЗначений Цикл
    НоваяСтрока = СтрокиДерева.Добавить();
    ЗаполнитьЗначенияСвойств(НоваяСтрока, Стр);
    КонецЦикла;

    Возврат
    КонецЕсли;

    ИмяГруппировки = Группировки[ИндексГруппировки];

    ЗначенияГруппировки = ТаблицаЗначений.Скопировать(, ИмяГруппировки);
    ЗначенияГруппировки.Свернуть(ИмяГруппировки);

    Для Каждого СтрГруппа Из ЗначенияГруппировки Цикл

    Значение = СтрГруппа[ИмяГруппировки];

    НоваяСтрокаДерева = СтрокиДерева.Добавить();
    Если СтрокиДерева.Родитель <> Неопределено Тогда
    ЗаполнитьЗначенияСвойств(НоваяСтрокаДерева, СтрокиДерева.Родитель);
    КонецЕсли;
    НоваяСтрокаДерева[ИмяГруппировки] = Значение;

    СтрокиГруппы = ТаблицаЗначений.Скопировать(Новый Структура(ИмяГруппировки, Значение));

    СоздатьСтрокиДереваРекурсивно(СтрокиГруппы, НоваяСтрокаДерева.Строки, Группировки, ИндексГруппировки + 1);

    КонецЦикла;
    КонецПроцедуры
```
---
Как же так? В таблицу ты смог добавить строки и заполнить их, а в дерево не можешь добавить и заполнить? Методы же одни и те же!
---
Вопрос именно в том чтобы заполнить не с нуля а из таблицы значения
---
Сообщение было скрыто модератором.
---
Сообщение было скрыто модератором.
---
Сообщение было скрыто модератором.
---
научимДа можно и без обучения его. Сразу.
---
научим
Лежать в пакетах? =))
---
Лень - двигатель прогресса...

```1c
    ТЗ = Новый ТаблицаЗначений();
    ТЗ.Колонки.Добавить("Товар", Новый ОписаниеТипов("Строка", , Новый КвалификаторыСтроки(50)));
    ТЗ.Колонки.Добавить("Количество", Новый ОписаниеТипов("Число", Новый КвалификаторыЧисла(1,0)));
    ТЗ.Колонки.Добавить("НаДату", Новый ОписаниеТипов("Дата"));

    Для Счетчик = 1 По 5 Цикл
    НовСтрока = ТЗ.Добавить();
    НовСтрока.Товар = "Товар " + Счетчик;
    НовСтрока.Количество = 10 * Счетчик;
    НовСтрока.НаДату = ТекущаяДата();
    КонецЦикла;

    ДеревоМечтаний = Новый ДеревоЗначений();

    Для каждого Колонка Из ТЗ.Колонки Цикл
    ДеревоМечтаний.Колонки.Добавить(Колонка.Имя, Колонка.ТипЗначения)
    КонецЦикла;

    Для каждого Строка Из ТЗ Цикл
    НовСтрока = ДеревоМечтаний.Строки.Добавить();
    ЗаполнитьЗначенияСвойств(НовСтрока, Строка);
    КонецЦикла;
```
---
Я решил задачу, вопрос было только понять синтаксис уровней как задается, почему то некоторые люди неадекватно начинают отвечать имея опыт за плечами.
---
Синтаксис строк вообще ничем не отличается в зависимости от уровней. Вопрос только в том, в какой уровень ты будешь засовывать свою строку, и по какому алгоритму определять эти уровни. А это никаким синтаксисом не помочь, тут только своими мозгами по своим критериям поиска/подбора и умением пользоваться самим этим поиском по дереву.
---
Каков вопрос -такой ответ. Ванговать на условия задачи - тратить время.
К слову, мой код был проверен в консоли все корректно. Так что учитесь ставить ТЗ.
Сээээр.
---
Итак...

1) Параметры не передаются

```1c
    Процедура ТаблицаЗначенийВДерево()
```

2) Заполнение колонок Дерева не из ТЗ а прямым перебором
3) Заполнение значений строки тоже по каждому полю. См. ЗаполнитьЗначенияСвойств

Вердикт: частное решение.

ЗЫ. моветон самому себе ставить решение, а еще выкладывать скринами, а не текстом.